Solving Simultaneous Difference Eqns.

TABLE OF CONTENTS


```
(%i5) load(draw)$ set_draw_defaults(line_width=2, grid = [2,2], point_type = filled_circle,
        head_type = 'nofilled, head_angle = 20, head_length = 0.5,
         background_color = light_gray, draw_realpart=false)$
        fpprintprec:5$ ratprint:false$ kill(all)$
```

```
(%i1) load ("Econ2.mac");
(%o1) c:/work5/Econ2.mac
```

1 Preface

Dowling19B.wxmx is one of a number of wxMaxima files available in the section
Economic Analysis with Maxima
on my CSULB webpage.

In Dowling19B.wxmx, we use Maxima to discuss the matrix solution of a set of n linear first order difference equations. We end with a reconsideration of the inflation and unemployment model discussed in Dowling18C.wxmx, following Chiang and Wainwright's Ch. 19, Sec. 4.

We have changed some of the symbols used in particular problems. An approximate pdf translation (using Microsoft print to pdf) is available as Dowling19Bfit.pdf. That pdf file can be searched using Ctrl-F.

Ted Woollett
https://home.csulb.edu/~woollett/
woollett@charter.net
Mar. 9, 2022

2 References

Introduction to Mathematical Economics, 3rd ed., Edward T. Dowling, 2012, Schaum's Outline Series, McGraw-Hill.

Fundamental Methods of Mathematical Economics, Alpha C. Chiang and Kevin Wainwright, 4th ed., 2005, McGraw-Hill

3 Matrix Solution of Y[t] = A . Y[t-1] + B

We have provided a brief review of Maxima matrix functions in Dowling19A.wxmx, which you should consult first.

If we let I represent the n x n unit matrix, each component of Ys[t] will be the matrix sum of the particular solution, which applies to a hypothetical time in the future when Y[t] = Y[t - 1] and the matrix equation becomes Yp = A . Yp + B, or (I - A) . Yp = B, whence

    Yp = invert(I - A) . B,

and Yc, the solution of the complementary equation Y[t] = A . Y[t - 1], which solution will have the form (if we assume distinct real eigenvalues)

    Yc = sum (k[j]*V[j]*r[j]^t, j, 1, n),

where k[j] are adjustable constants, and r[j] and V[j] are respectively eigenvalues and corresponding eigenvectors must satisfy the equations (j = 1 thru n)

1. determinant (A - r[j]*I) = 0
2. A . V[j] = r[j]*V[j]

The constants k[j] in Yc are determined by using prescribed initial values Y[0]. Again assuming distinct real eigenvalues, the dominant root is the root with the largest absolute value. For convergence we need the magnitude of the dominant root to be less than 1.

3.1 Proof of the Yc Solution Form for n = 2

For n = 2 we assert Yc[t] has the form

    Yc[t] = k1*r1^t*V1 + k2*r2^t*V2.

We need to show that Yc[t] = A . Yc[t-1].

Now A . Yc[t-1] = k1*r1^(t-1)* (A . V1) + k2*r2^(t-1)* (A . V2)
= k1*r1^(t-1)*r1*V1 + k2*r2^(t-1)*r2*V2
= k1*r1^t*V1 + k2*r2^t*V2 = Yc[t].

3.2 Example 5: n = 2 Distinct Real Roots

Solve the following system of first-order linear difference equations:

    x[t] = - 4*x[t-1] + y[t-1] + 12, x[0] = 16,
    y[t] =   2*x[t-1] - 3*y[t-1] + 6, y[0] = 8.

3.2.1 Using Ytlist (A, B, Y0, tmax)

The quickest way to produce a visual look at the solution of a set of first order difference equations written in the matrix form
  Y[t] = A . Y[t-1] + B
is to use the Maxima function (defined in Econ2.mac) Ytlist (A, B, Y0, tmax).

Ytlist (A, B, Y0, tmax) produces a column vector whose jth component is the list of values
    [yj[0], yj[1], yj[2],...,yj[tmax]],
using directly the first order difference equation Y[t] = A . Y[t-1] + B repeatedly, starting with Y[1] = A . Y0 + B, without using the eigenvalues or eigenvectors of A.

We revert to the notation Ye (instead of Yp) for the particular solution, since its physical meaning is the (hypothetical) long time equilibrium value of the system vector Y[t].

```
(%i7) A : matrix ( [-4, 1], [2, -3] );
      eigenvalues (A);
      B : cvec ( [12, 6]);
      Y0 : cvec ([16,8] );
      Ye : invert (ident(2) - A) . B;
      Yt : Ytlist (A, B, Y0, 5);
```

(A) $\begin{pmatrix} -4 & 1 \\ 2 & -3 \end{pmatrix}$

(%o3) $[[-5,-2],[1,1]]$

(B) $\begin{pmatrix} 12 \\ 6 \end{pmatrix}$

(Y0) $\begin{pmatrix} 16 \\ 8 \end{pmatrix}$

(Ye) $\begin{pmatrix} 3 \\ 3 \end{pmatrix}$

(Yt) $\begin{pmatrix} [16,-44,202,-920,4474,-22064] \\ [8,14,-124,782,-4180,21494] \end{pmatrix}$

```
(%i8) grind (Yt)$
matrix([[16,-44,202,-920,4474,-22064]],[[8,14,-124,782,-4180,21494]])$
```

The dominant root is -5 whose magnitude is greater than 1, hence a divergent system.

Let xs be the list of values of x[t] from t = 0 to t = 5. Likewise, ys is the list of values of y[t].

lme is our alias (defined in Econ2.mac) for list_matrix_entries.

```
(%i9)  [xs, ys] : lme (Yt);
(%o9)  [[16,−44,202,−920,4474,−22064],[8,14,−124,782,−4180,
       21494]]

(%i10) xs;
(%o10) [16,−44,202,−920,4474,−22064]
```

The list of times tL should be the same length as xs or ys.

```
(%i11) tL : makelist (t, t, 0, 5);
(tL)   [0,1,2,3,4,5]
```

We are only going to display the values of x[t] and y[t] out to the time t = 3 on this plot.

```
(%i12) wxdraw2d (xlabel = "t", xrange = [0, 3.5], yrange = [-1e3, 1e3],
         key_pos = top_left,  key = "x[t]", points (tL, xs), color = red, key = "y[t]",
         points (tL, ys), color = black, line_width = 1, key = "",
         explicit (0, t, 0,3.5))$
```

(%t12)



We see the expected divergent behavior of this system.

3.2.2 Using DE1S (A, B, t, Y0)

The Maxima function DE1S (A, B, t, Y0), defined in Econ2.mac, uses the eigenvalues and corresponding eigenvectors of the square matrix A to find a solution to the complementary equation Yc[t] = A . Yc[t-1], and that solution has the form
Yc = k1*r1^t*V1 + k2*r2^t*V2 (k1 and k2 being arbitrary constants)
(if the eigenvalues are real and distinct) for n = 2, where V1 and V2 are the corresponding eigenvectors: A . V1 = r1*V1, A . V2 = r2*V2.

The indefinite solution is then Yc + Ye, where Ye is the hypothetical long time equilibrium state in which Y[t] = Y[t-1], so Ye = A . Ye + B, and hence
Ye = invert (I - A) . B,
where I = ident(n) is the n x n unit matrix.

DE1S (A, B, t, Y0) or DE1S (A, B, t) returns Y[t], the matrix column vector solution
of a set of n first order difference equations expressed in matrix form as
Y[t] = A . Y[t-1] + B.
Y0 is the optional matrix column vector of initial values.
Y[t] has with n components if A is a square n x n matrix, and is returned in
definite form if the fourth arg Y0 is included, else in terms of n constants _k[j].

(%i13) DE1S (A, B, t);

(%o13) $$\begin{pmatrix} \_k_2\,(-2)^t+\_k_1\,(-5)^t+3 \\ -\_k_2\,(-2)^{t+1}-\_k_1\,(-5)^t+3 \end{pmatrix}$$

(%i14) Ys : DE1S (A, B, t, Y0);

(Ys) $$\begin{pmatrix} -3\,(-2)^{t+1}+7\,(-5)^t+3 \\ 3\,(-2)^{t+2}-7\,(-5)^t+3 \end{pmatrix}$$

Because we have included Y0, the definite form is returned which is written using the eigenvalues -2 and -5 (call them r1 and r2 for this example)

(%i15) [xs, ys] : lme (Ys);

(%o15) $[-3\,(-2)^{t+1}+7\,(-5)^t+3, 3\,(-2)^{t+2}-7\,(-5)^t+3]$

(%i16) xs;

(%o16) $-3\,(-2)^{t+1}+7\,(-5)^t+3$

Here, xs and ys are not lists, rather they are each a Maxima expression depending on the parameter t.

(%i17) at ([xs, ys], t = 0);

(%o17) $[16, 8]$

(%i18) txL : makelist ([tt, at (xs, t = tt)], tt, 0, 6);

(txL) [[0,16],[1,−44],[2,202],[3,−920],[4,4474],[5,−22064],[6,109762]]

(%i19) tyL : makelist ([tt, at (ys, t = tt)], tt, 0, 6);

(tyL) [[0,8],[1,14],[2,−124],[3,782],[4,−4180],[5,21494],[6,−108604]]

The divergence is clear from these values of x[t] and y[t].

We again only look out to t = 3.

(%i20) wxdraw2d (xlabel = "t", xrange = [0, 3.5], yrange = [-1e3, 1e3],
key_pos = top_left, key = "x[t]", points (txL), color = red, key = "y[t]",
points (tyL), color = black, line_width = 1, key = "",
explicit (0, t, 0,3.5))$

(%t20)



This plot reproduces the plot made above using Ytlist.

3.2.3 Matrix Details of DE1S Code

The solution Yc of the complementary equation
Yc[t] = A . Yc[t-1]
depends on the eigenvalues and corresponding eigenvectors of A.

(%i21) [eval, evec] : eigenvectors (A);

(%o21) [[[−5,−2],[1,1]],[[[1,−1]],[[1,2]]]]

(%i22) evec;

(%o22) [[[1,−1]],[[1,2]]]

(%i23) evec[1][1];

(%o23) [1,−1]

(%i24) evec[2][1];

(%o24) [1,2]

(%i25) eval;

(%o25) [[−5,−2],[1,1]]

(%i26) eval[1];

(%o26) [−5,−2]

(%i27) eval[1][1];

(%o27) −5

The eigenvalues of A are real and distinct and each have magnitudes greater than 1, which leads to a divergent system time path.

```
(%i33) r1 : -5;
       V1 : cvec ([1, -1]);
       A . V1;
       r2 : -2;
       V2 : cvec ([1,2]);
       A . V2;
```

(r1) −5

(V1) $\begin{pmatrix} 1 \\ -1 \end{pmatrix}$

(%o30) $\begin{pmatrix} -5 \\ 5 \end{pmatrix}$

(r2) −2

(V2) $\begin{pmatrix} 1 \\ 2 \end{pmatrix}$

(%o33) $\begin{pmatrix} -2 \\ -4 \end{pmatrix}$

Here is the complementary equation solution Yc.

(%i34) Yc : k1*r1^t*V1 + k2*r2^t*V2;

(Yc) $\begin{pmatrix} k2\,(-2)^t + k1\,(-5)^t \\ -k2\,(-2)^{t+1} - k1\,(-5)^t \end{pmatrix}$

Adding up the matrix column vectors Yc and Ye (defined way up above) defines the "indefinite solution" Yindef:

(%i35) Yindef : Yc + Ye;

(Yindef) $\begin{pmatrix} k2\,(-2)^t + k1\,(-5)^t + 3 \\ -k2\,(-2)^{t+1} - k1\,(-5)^t + 3 \end{pmatrix}$

The conponents of Y0 are x[0] and y[0]. (We defined this way above).

(%i36) Y0 ;

(%o36) $\begin{pmatrix} 16 \\ 8 \end{pmatrix}$

We use our Maxima function colVecSolve (col1, col2), defined in Econ2.mac, to require Yindef = Y0 when t = 0.

(%i37) solns : colVecSolve (at (Yindef, t = 0), Y0);

(solns) $[k2 = 6, k1 = 7]$

Let Ys be our solution Y[t] taking into account our initial conditions.

(%i38) Ys : at (Yindef, solns);

(Ys) $\begin{pmatrix} -3\,(-2)^{t+1} + 7\,(-5)^t + 3 \\ 3\,(-2)^{t+2} - 7\,(-5)^t + 3 \end{pmatrix}$

(%i39) grind (Ys)$

matrix([(-3*(-2)^(t+1))+7*(-5)^t+3],[3*(-2)^(t+2)-7*(-5)^t+3])$

Let xs and ys be the components of Ys. lme is our alias for list_matrix_entries and is defined in Econ2.mac.

(%i40) [xs, ys] : lme (Ys);

(%o40) $[-3\,(-2)^{t+1} + 7\,(-5)^t + 3, 3\,(-2)^{t+2} - 7\,(-5)^t + 3]$

(%i41) at ([xs, ys], t = 0);

(%o41) [16,8]

(%i42) txL : makelist ([tt, at (xs, t = tt)],tt, 0, 6);

(txL) [[0,16],[1,−44],[2,202],[3,−920],[4,4474],[5,−22064],[6,
109762]]

(%i43) tyL : makelist ([tt, at (ys, t = tt)],tt, 0, 6);

(tyL) [[0,8],[1,14],[2,−124],[3,782],[4,−4180],[5,21494],[6,−
108604]]

The divergence is clear from these values of x[t] and y[t].

We can use points (txL) , for example, to show the values of x[t] on a plot of x[t] versus t, for discrete values of t = 0, 1, 2, ...

(%i44) wxdraw2d (xlabel = "t", xrange = [0, 3.5], yrange = [-1e3, 1e3],
key_pos = top_left, key = "x[t]", points (txL), color = red, key = "y[t]",
points (tyL), color = black, line_width = 1, key = "",
explicit (0, t, 0,3.5))$

(%t44)



4 Example with n = 2 Complex Roots

(%i47) kill (a,b,c,d)$
A : matrix ([a,b],[c,d]);
eigenvalues (A);

(A) $\begin{pmatrix} a & b \\ c & d \end{pmatrix}$

(%o47) $[[-\frac{\sqrt{d^2-2\,a\,d+4\,b\,c+a^2}-d-a}{2},\frac{\sqrt{d^2-2\,a\,d+4\,b\,c+a^2}+d+a}{2}],[1,1]]$

To have complex eigenvalues we need the arg of the square root equal to a negative number, or (d - a)^2 + 4*b*c < 0, or (d - a)^2 < - 4*b*c, so we need b # 0 and c # 0 and b and c to have opposite signs. If we take c = - e*b, then we need e > 0 and b # 0. Example: take c = - b, then we need (d - a)^2 < 4*b^2. If we take b = 2, we need (d - a)^2 < 16, which is satisfied if we take a = 2, d = 4, d - a = 2, (d - a)^2 = 4 < 16.

(%i49) A : matrix ([2, 2], [-2, 4]);
[r1, r2] : eigenvalues (A)[1];

(A) $\begin{pmatrix} 2 & 2 \\ -2 & 4 \end{pmatrix}$

(%o49) $[3-\sqrt{3}\,\%i,\sqrt{3}\,\%i+3]$

(%i50) map ('abs, [r1, r2]);

(%o50) $[2\sqrt{3},2\sqrt{3}]$

(%i51) float(%);

(%o51) $[3.4641,3.4641]$

Since we can write, using |r1| = sqrt (x1^2 + y1^2) if r1 = x1 + %i*y1,
r1 = |r1|*exp (%i*th1) = |r1|* (sin (th1) + %i*cos (th1)), and when we raise r1 to the integral power t, we get
r1^t = |r1|^t*exp (%i*t*th1)
and since the part of Yc proportional to r1^t is
k1*r1^t*V1 = k1*|r1|^t*exp (%i*t*th1)*V1,
we need |r1| < 1 for this part of Yc to converge. Since r1 and r2 will be complex conjugates of each other, we need the absolute value of either root to be less than 1 for convergence.

Maxima's abs or cabs functions will return the absolute value of a number,
where if r = x + %i*y, |r| = sqrt (x^2 + y^2).

(%i52) abs (3 - %i*sqrt(3));

(%o52) $2\sqrt{3}$

(%i53) float(%);

(%o53) 3.4641

Since this is greater than 1, the solution is divergent.

We take the same B and Y0.

(%i55) B;
Y0;

(%o54) $\begin{pmatrix} 12 \\ 6 \end{pmatrix}$

(%o55) $\begin{pmatrix} 16 \\ 8 \end{pmatrix}$

(%i56) Ye : invert (ident(2) - A) . B, numer;

(Ye) $\begin{pmatrix} -3.4286 \\ -4.2857 \end{pmatrix}$

4.1 Using Ytlist (A, B, Y0, tmax)

(%i57) Yt : Ytlist (A, B, Y0, 6);

(Yt) $\begin{pmatrix} [16,60,144,120,-1032,-7656,-33576] \\ [8,6,-90,-642,-2802,-9138,-21234] \end{pmatrix}$

(%i58) [xs, ys] : lme (Yt);

(%o58) [[16,60,144,120,−1032,−7656,−33576],[8,6,−90,−642,−2802,
−9138,−21234]]

(%i59) xs;

(%o59) [16,60,144,120,−1032,−7656,−33576]

(%i60) txL : makelist ([t, xs[t+1]], t, 0, 4);

(txL) [[0,16],[1,60],[2,144],[3,120],[4,−1032]]

(%i61) tyL : makelist ([t, ys[t+1]], t, 0, 4);

(tyL) [[0,8],[1,6],[2,−90],[3,−642],[4,−2802]]

(%i62) wxdraw2d (xlabel = "t", xrange = [0, 4.5], yrange = [-3e3, 300],
key_pos = bottom_left, key = "x[t]", points (txL), color = red, key = "y[t]",
points (tyL), color = black, line_width = 1, key = "",
explicit (0, t, 0, 4.5))$

(%t62)



4.2 Using DE1S (A, B, t, Y0)

(%i63) Ys : DE1S (A, B, t, Y0);

(Ys)

$$\begin{pmatrix} \frac{(3-\sqrt{3}\,\%i)^t\,(2\,3^{3/2}\,\%i+68)}{7}-\frac{(\sqrt{3}\,\%i+3)^t\,(2\,3^{3/2}\,\%i-68)}{7}-\frac{24}{7} \\ -\frac{(3-\sqrt{3}\,\%i)^t\,(\sqrt{3}\,\%i-1)\,(2\,3^{3/2}\,\%i+68)}{14}-\frac{(\sqrt{3}\,\%i+1)\,(\sqrt{3}\,\%i+3)^t\,(2\,3^{3/2}\,\%i-68)}{14}-\frac{30}{7} \end{pmatrix}$$

Because we have included Y0, the definite form is returned which is written using the eigenvalues (call them r1 and r2 for this example)

(%i64) [xs, ys] : lme (Ys);

(%o64)

$$\left[\frac{(3-\sqrt{3}\,\%i)^t\,(2\,3^{3/2}\,\%i+68)}{7}-\frac{(\sqrt{3}\,\%i+3)^t\,(2\,3^{3/2}\,\%i-68)}{7}-\frac{24}{7},-\frac{(3-\sqrt{3}\,\%i)^t\,(\sqrt{3}\,\%i-1)\,(2\,3^{3/2}\,\%i+68)}{14}-\frac{(\sqrt{3}\,\%i+1)\,(\sqrt{3}\,\%i+3)^t\,(2\,3^{3/2}\,\%i-68)}{14}-\frac{30}{7}\right]$$

(%i65) xs;

(%o65) $$\frac{(3-\sqrt{3}\,\%i)^t\,(2\,3^{3/2}\,\%i+68)}{7}-\frac{(\sqrt{3}\,\%i+3)^t\,(2\,3^{3/2}\,\%i-68)}{7}-\frac{24}{7}$$

(%i66) at ([xs, ys], t = 0), ratsimp;

(%o66) [16,8]

(%i67) txL : makelist ([tt, ratsimp (at (xs, t = tt))], tt, 0, 6);

(txL) [[0,16],[1,60],[2,144],[3,120],[4,−1032],[5,−7656],[6,−33576]]

(%i68) tyL : makelist ([tt, ratsimp (at (ys, t = tt))], tt, 0, 6);

(tyL) [[0,8],[1,6],[2,−90],[3,−642],[4,−2802],[5,−9138],[6,−21234]]

(%i69) wxdraw2d (xlabel = "t", xrange = [0, 4.5], yrange = [-3e3, 300],
key_pos = bottom_left, key = "x[t]", points (txL), color = red, key = "y[t]",
points (tyL), color = black, line_width = 1, key = "",
explicit (0, t, 0, 4.5))$

(%t69)



4.3 n = 2 Complex Root Case Convergence

Begin with the form of the matrix A developed above to produce complex roots.
A : matrix ([a, b], [-e*b, d])
with e > 0 and b # 0. What must be true to now have convergence?

(%i71) A : matrix ([a, b], [- e*b, d]);
[r1, r2] : eigenvalues (A)[1];

(A) $\begin{pmatrix} a & b \\ -b\,e & d \end{pmatrix}$

(%o71) $[-\frac{\sqrt{-4\,b^2\,e+d^2-2\,a\,d+a^2}-d-a}{2},\frac{\sqrt{-4\,b^2\,e+d^2-2\,a\,d+a^2}+d+a}{2}]$

Assume 4*e*b^2 > (d - a)^2 to have complex roots, and rewrite r2 as
r2 = (1/2)*(d + a) + %i*(1/2)*sqrt(4*e*b^2 - (d-a)^2) = x2 + %i*y2.
Then |r2|^2 = x2^2 + y2^2:

(%i72) expand ((d + a)^2/4 + (4*e*b^2 - (d - a)^2)/4);

(%o72) $b^2\,e+a\,d$

Thus |r2| = sqrt (a*d + e*b^2), and e > 0, b^2 > 0.
Thus for convergence we need (a*d + e*b^2) < 1.

With a = b = d = 1/2 and e = 1 we then expect convergence.

(%i74) A : matrix ([1/2, 1/2], [- 1/2, 1/2]);
[r1, r2] : eigenvalues (A)[1];

(A) $\begin{pmatrix} \frac{1}{2} & \frac{1}{2} \\ -\frac{1}{2} & \frac{1}{2} \end{pmatrix}$

(%o74) $[-\frac{\%i-1}{2},\frac{\%i+1}{2}]$

(%i75) map ('abs, [r1, r2]);

(%o75) $[\frac{1}{\sqrt{2}},\frac{1}{\sqrt{2}}]$

(%i76) float (%);

(%o76) [0.70711,0.70711]

Since the magnitude of the complex roots is ~ 0.71 < 1, we expect convergence.

Again we assume the same B and Y0.

(%i78) B;
Y0;

(%o77) $\begin{pmatrix}12\\6\end{pmatrix}$

(%o78) $\begin{pmatrix}16\\8\end{pmatrix}$

(%i79) Ye : invert (ident(2) - A) . B, numer;

(Ye) $\begin{pmatrix}18.0\\-6.0\end{pmatrix}$

(%i80) Yt : Ytlist (A, B, Y0, 8);

(Yt) $\begin{pmatrix}[16,24,25,22,\frac{37}{2},\frac{33}{2},\frac{65}{4},17,\frac{143}{8}]\\[8,2,-5,-9,-\frac{19}{2},-8,-\frac{25}{4},-\frac{21}{4},-\frac{41}{8}]\end{pmatrix}$

(%i81) [xs, ys] : lme (Yt);

(%o81) $[[16,24,25,22,\frac{37}{2},\frac{33}{2},\frac{65}{4},17,\frac{143}{8}],[8,2,-5,-9,-\frac{19}{2},-8,-\frac{25}{4},-\frac{21}{4},-\frac{41}{8}]]$

(%i82) xs,numer;

(%o82) [16,24,25,22,18.5,16.5,16.25,17,17.875]

(%i83) ys,numer;

(%o83) [8,2,-5,-9,-9.5,-8,-6.25,-5.25,-5.125]

(%i84) txL : makelist ([t, xs[t+1]], t, 0, 8);

(txL) $[[0,16],[1,24],[2,25],[3,22],[4,\frac{37}{2}],[5,\frac{33}{2}],[6,\frac{65}{4}],[7,17],[8,\frac{143}{8}]]$

(%i85) tyL : makelist ([t, ys[t+1]], t, 0, 8);

(tyL) $[[0,8],[1,2],[2,-5],[3,-9],[4,-\frac{19}{2}],[5,-8],[6,-\frac{25}{4}],[7,-\frac{21}{4}],[8,-\frac{41}{8}]]$

(%i87) tmax : 8.5$
wxdraw2d (xlabel = "t", xrange = [0, tmax], yrange = [-30, 30],
key_pos = bottom_left, key = "x[t]", points (txL), color = red, key = "y[t]",
points (tyL), line_width = 1, color = blue, key = "xe = 18", explicit (18, t, 0, tmax),
color = red, key = "ye = - 6", explicit (-6, t, 0, tmax), color = black, key = "",
explicit (0, t, 0, tmax))$

(%t87)



We see convergence (with oscillation) in this complex root example.

5 n = 2 Repeated Real Roots Case

(%i90) kill (a, b, c, d)$
A : matrix ([a, b], [c, d]);
[r1, r2] : eigenvalues (A)[1];

(A) $\begin{pmatrix} a & b \\ c & d \end{pmatrix}$

(%o90) $[-\frac{\sqrt{d^2-2\,a\,d+4\,b\,c+a^2}-d-a}{2}, \frac{\sqrt{d^2-2\,a\,d+4\,b\,c+a^2}+d+a}{2}]$

What do we need to get r1 = r2?

(%i91) r1mr2 : expand (2*r1 - 2*r2);

(r1mr2) $-2\sqrt{d^2-2\,a\,d+4\,b\,c+a^2}$

(%i92) r1mr2 : r1mr2/(-2);

(r1mr2) $\sqrt{d^2-2\,a\,d+4\,b\,c+a^2}$

(%i93) r1mr2 : r1mr2^2;

(r1mr2) $d^2-2\,a\,d+4\,b\,c+a^2$

So for two equal roots we need (d-a)^2 + 4*b*c = 0, so the sign of c must be opposite to the sign of b, so replace c by -e so A = matrix ([a,b],[-e,d]), and then we need 4*b*e = (d-a)^2, or A = matrix ([a, b], [-(d-a)^2/(4*b), d]).

(%i95) A : matrix ([a, b], [- (d-a)^2/(4*b), d]);
eigenvalues (A);

(A) $\begin{pmatrix} a & b \\ -\frac{(d-a)^2}{4\,b} & d \end{pmatrix}$

(%o95) $[[\frac{d+a}{2}],[2]]$

So this is the case we want, one root r = (d+a)/2 with multiplicity 2, and we can assume a solution of the form
Yc[t] = (k1*r^t + k2*r^(t + 1))*V, where A . V = r*V,
since Yc[t-1] = (k1*r^(t-1) + k2*r^t) * V, and then A . Yc[t-1] = Yc[t].

Here is an example which results in a repeated real root r = 1/2, with |r| < 1, and a convergent system. We continue to use the same matrix column vectors for B and Y0 as above. We find a brute force solution by using Ytlist(A,B,Y0, tmax).

(%i97) A : matrix ([1/2, 1/2], [0, 1/2]);
eigenvalues (A);

(A) $\begin{pmatrix} \frac{1}{2} & \frac{1}{2} \\ 0 & \frac{1}{2} \end{pmatrix}$

(%o97) $[[\frac{1}{2}],[2]]$

(%i99) B;
Y0;

(%o98) $\begin{pmatrix} 12 \\ 6 \end{pmatrix}$

(%o99) $\begin{pmatrix} 16 \\ 8 \end{pmatrix}$

(%i100) Ye : invert (ident(2) - A) . B, numer;

(Ye) $\begin{pmatrix} 36.0 \\ 12.0 \end{pmatrix}$

(%i101) Yt : Ytlist (A, B, Y0, 8), numer;

(Yt) $\begin{pmatrix} [16,24.0,29.0,32.0,33.75,34.75,35.313,35.625,35.797] \\ [8,10.0,11.0,11.5,11.75,11.875,11.938,11.969,11.984] \end{pmatrix}$

(%i102) [xs, ys] : lme (Yt);

(%o102) [[16,24.0,29.0,32.0,33.75,34.75,35.313,35.625,35.797],[8,10.0,11.0,11.5,11.75,11.875,11.938,11.969,11.984]]

(%i103) xs;

(%o103) [16,24.0,29.0,32.0,33.75,34.75,35.313,35.625,35.797]

(%i104) txL : makelist ([t, xs[t+1]], t, 0, 8);

(txL) [[0,16],[1,24.0],[2,29.0],[3,32.0],[4,33.75],[5,34.75],[6,35.313],[7,35.625],[8,35.797]]

(%i105) tyL : makelist ([t, ys[t+1]], t, 0, 8);

(tyL) [[0,8],[1,10.0],[2,11.0],[3,11.5],[4,11.75],[5,11.875],[6,11.938],[7,11.969],[8,11.984]]

```
(%i107) tmax : 8.5$
        wxdraw2d (xlabel = "t", xrange = [0, tmax], yrange = [0, 70],
          key_pos = top_left,  key = "x[t]", points (txL),  color = red, key = "y[t]",
          points (tyL), color = blue, line_width = 1, key = "xe = 36",
          explicit (36, t, 0, tmax), color = red, key =  "ye = 12",
          explicit (12, t, 0, tmax))$
```

(%t107)




```
(%i108) DE1S (A, B, t, Y0);
(%o108) DE1S is not able to handle repeated roots
```

6 *Matrix Solution of A1 . Y[t] = A2 . Y[t-1] + B*

```
(%i109) DE2S (_A1, _A2, _B, _t, _Y0) :=
        block (
          if det (_A1) = 0 then return (" DE2S method depends on det (A1) not being zero"),
          DE1S (invert (_A1) . _A2, invert (_A1) . _B, _t, _Y0))$
```

6.1 Dowling Ex. 7 & 8

Solve the set of first order difference equations
  x[t] = 4*x[t-1] - 2*y[t-1] + y[t] - 10, x[0] = 20,
  y[t] = 3*x[t-1] + 6*y[t-1] - 4, y[0] = 3.

Writing this set in matrix form
  A1 . Y[t] = A2 . Y[t-1] + B,
with the initial values in Y0,

```
(%i113) A1 : matrix ([1,-1], [0, 1]);
        A2 : matrix ([4, -2], [3, 6]);
        B : cvec ([-10, -4]);
        Y0 : cvec ([20, 3]);
```

(A1) $\begin{pmatrix} 1 & -1 \\ 0 & 1 \end{pmatrix}$

(A2) $\begin{pmatrix} 4 & -2 \\ 3 & 6 \end{pmatrix}$

(B) $\begin{pmatrix} -10 \\ -4 \end{pmatrix}$

(Y0) $\begin{pmatrix} 20 \\ 3 \end{pmatrix}$

6.1.1 Using Ytlist (D, E, Y0, tmax)

We can convert the given matrix equation to the form
Y[t] = D . Y[t-1] + E, with Y0 unchanged, by multiplying the given matrix equation from the left by invert(A1), and using invert (A1) . A1 = I = unit matrix and then using I . Y[t] = Y[t], and defining D and E matrices:

```
(%i115) D : invert (A1) . A2;
        E : invert (A1) . B;
```

(D) $\begin{pmatrix} 7 & 4 \\ 3 & 6 \end{pmatrix}$

(E) $\begin{pmatrix} -14 \\ -4 \end{pmatrix}$

```
(%i116) eigenvalues (D);
(%o116) [[3,10],[1,1]]
```

Real eigenvalues with magnitudes much greater than 1 implies a very divergent system.

```
(%i118) Ye : invert (ident(2) - D) . E;
        Ys : Ytlist (D, E, Y0, 6);
```

(Ye) $\begin{pmatrix} 3 \\ -1 \end{pmatrix}$

(Ys) $\begin{pmatrix} [20,138,1248,12138,120408,1201218,12003648] \\ [3,74,854,8864,89594,898784,8996354] \end{pmatrix}$

(%i119) [xs, ys] : lme (Ys);

(%o119) [[20,138,1248,12138,120408,1201218,12003648],[3,74,854,8864,89594,898784,8996354]]

(%i120) tL : makelist (tt, tt, 0, 6);

(tL) [0,1,2,3,4,5,6]

(%i122) tmax : 2.5$
wxdraw2d (xlabel = "t", xrange = [0, tmax], yrange = [0, 1.5e3],
key_pos = top_left, key = "x[t]", points (tL, xs), color = red, key = "y[t]",
points (tL, ys))$

(%t122) 

6.1.2 Using DE2S (A1, A2, B, t, Y0)

(%i123) DE2S (A1,A2,B, t,Y0);

(%o123) $$\begin{pmatrix} 12\ 10^t+5\ 3^t+3 \\ 9\ 10^t-5\ 3^t-1 \end{pmatrix}$$

7 *Inflation-Unemployment Set of 1st Order Difference Eqns*

We follow Chaing and Wainwright's Ch. 19 treatment of this topic. In Dowling18C.wxmx we arrived at a set of discrete time equations, which we conbined into one second order difference equation. Here we want to work with a set of two first order difference equations.

With p the actual inflation rate (%), π the expected inflation rate (%), and U the unemployment rate (%), μ the fixed rate of increase of money [μ = (1/M)*dM/dt] set as a monetary policy decision by the FED, in 18C we translated the continuous time equations into a discrete time version (here we have replaced m --> μ):

p[t] = α - T - β*U[t] + g*π[t], (0 < g <= 1), (α, β > 0),

π[t+1] - π[t] = j*(p[t] - π[t]), (0 < j <= 1), and

U[t+1] - U[t] = -k*(μ - p[t+1]), (k > 0).

We eliminate p[t] and p[t+1] in the second and third equations:
π[t+1] - π[t] = j*(α - T - β*U[t] + g*π[t]) - j*π[t], (1)

U[t+1] - U[t] = -k*μ + k* (α - T - β*U[t+1] + g*π[t+1]) (2)

Our methods in this chapter 19B are based on the two matrix forms
Y[t] = A . Y[t-1] + B, or
A1. Y[t] = A2 . Y[t-1] + B.

So we replace t --> t - 1 in both (1) and (2) to get

π[t] = (1 - j*(1-g))*π[t-1] - j*β*U[t-1] + j*(α - T), (3)

- k*g*π[t] + (1 + β*k)*U[t] = U[t-1] + k*(α - T - μ). (4)

We can then write the pair of equations (3) and (4) in matrix form as

A1 . Y[t] = A2 . Y[t-1] + B (5)

with the following assignments:

```
(%i126) A1 : matrix ([1, 0], [-k*g, 1 + β*k]);
        A2 : matrix ([1 - j*(1 - g), - j*β], [0, 1]);
        B : cvec ([j*(α  - T), k*(α - T - μ)]);
```

(A1) $\begin{pmatrix} 1 & 0 \\ -g\,k & k\,\beta+1 \end{pmatrix}$

(A2) $\begin{pmatrix} 1-(1-g)\,j & -j\,\beta \\ 0 & 1 \end{pmatrix}$

(B) $\begin{pmatrix} j\,(\alpha-T) \\ k\,(-\mu+\alpha-T) \end{pmatrix}$

In Dowling18C.wxmx we considered two cases, which we will call case1 and case2 here. We defined a list of parameter replacements, but replace α - T = 11 separately using ratsubst.

7.1 Case 1: Real Eigenvalues

```
(%i127) case1 : [β = 2, k = 5/2, g = 1/2, j = 1/3, μ = 2];
```

(case1) $[\beta=2, k=\frac{5}{2}, g=\frac{1}{2}, j=\frac{1}{3}, \mu=2]$

```
(%i131) A1_case1 : at (A1, case1);
        A2_case1 : at (A2, case1);
        B_case1 : at (B, case1);
        B_case1 : ratsubst (11, α - T, B_case1);
```

(A1_case1) $\begin{pmatrix} 1 & 0 \\ -\frac{5}{4} & 6 \end{pmatrix}$

(A2_case1) $\begin{pmatrix} \frac{5}{6} & -\frac{2}{3} \\ 0 & 1 \end{pmatrix}$

(B_case1) $\begin{pmatrix} \frac{\alpha-T}{3} \\ \frac{5\,(\alpha-T-2)}{2} \end{pmatrix}$

(B_case1) $\begin{pmatrix} \frac{11}{3} \\ \frac{45}{2} \end{pmatrix}$

7.1.1 Using Ytlist (D, E, Y0, tmax)

One approach is to multiply the matrix equation from the left by invert (A1), arriving at the equation Y[t] = D . Y[t-1] + E, (using invert(A1) . A1 = ident(2) = 2 x 2 unit matrix, and if I is the appropriate unit matix, I . A = A).

```
(%i133) D : invert (A1_case1) . A2_case1;
        E : invert (A1_case1) . B_case1;
```

(D) $$\begin{pmatrix} \frac{5}{6} & -\frac{2}{3} \\ \frac{25}{144} & \frac{1}{36} \end{pmatrix}$$

(E) $$\begin{pmatrix} \frac{11}{3} \\ \frac{325}{72} \end{pmatrix}$$

```
(%i134) eigenvalues (D), numer;
(%o134) [[0.21494,0.64617],[1,1]]
```

We have real eigenvalues with magnitudes less than 1, so we expect eventual convergence.

```
(%i135) Ye : invert (ident(2) - D) . E;
```

(Ye) $$\begin{pmatrix} 2 \\ 5 \end{pmatrix}$$

So we expect long time convergence with πe = 2 and Ue = 5.

For our first plot, we assume π[0] = 4 and U[0] = 1.

```
(%i136) Y0 : cvec ([4, 1]);
```

(Y0) $$\begin{pmatrix} 4 \\ 1 \end{pmatrix}$$

```
(%i137) Ys : Ytlist (D, E, Y0, 6), numer;
```

(Ys) $$\begin{pmatrix} [4,6.3333,5.4537,4.3722,3.563,3.0165,2.6582] \\ [1,5.2361,5.7589,5.6207,5.4291,5.2833,5.1843] \end{pmatrix}$$

(%i138) [πs, Us] : lme (Ys);

(%o138) [[4,6.3333,5.4537,4.3722,3.563,3.0165,2.6582],[1,5.2361,5.7589,5.6207,5.4291,5.2833,5.1843]]

We need the list of times (tL) to be the same length as πs and Us.

(%i139) tL : makelist (tt, tt, 0, 6);

(tL) [0,1,2,3,4,5,6]

(%i140) wxdraw2d (xlabel = "t", yrange = [0,10], xrange = [0, 6],
title = "g = 1/2, j = 1/3, β = 2, k = 5/2, μ=2, α - T = 11",
key = "U[t]", points (tL, Us), color = magenta, key = "pi[t]",
points (tL, πs), color = black, line_width = 1, key = "Ue = 5",
explicit (5, t, 0, 6), color = brown, key = "μ = 2", explicit (2, t, 0, 6))$

(%t140)



7.1.2 Using DE1S (D, E, t, Y0)

(%i141) Ys : DE1S (D, E, t, Y0);

(Ys)

$$\begin{pmatrix} \frac{(\sqrt{241}+31)^t\,(125\sqrt{241}+241)}{241\;72^t} - \frac{(31-\sqrt{241})^t\,(125\sqrt{241}-241)}{241\;72^t} + 2 \\ -\frac{(\sqrt{241}-29)\,(\sqrt{241}+31)^t\,(125\sqrt{241}+241)}{11568\;72^t} - \frac{(31-\sqrt{241})^t\,(\sqrt{241}+29)\,(125\sqrt{241}-241)}{11568\;72^t} + 5 \end{pmatrix}$$

(%i142) [πs, Us] : lme (Ys),numer;

(%o142) $\left[\frac{9.052\ 46.524^t}{72^t}-\frac{7.052\ 15.476^t}{72^t}+2, \frac{2.5413\ 46.524^t}{72^t}-\frac{6.5413\ 15.476^t}{72^t}+5\right]$

(%i143) tπL : makelist ([tt, at (πs, t = tt)], tt, 0, 6);

(tπL) [[0,4.0],[1,6.3333],[2,5.4537],[3,4.3722],[4,3.563],[5,3.0165],[6,2.6582]]

(%i144) tUL : makelist ([tt, at (Us, t = tt)], tt, 0, 6);

(tUL) [[0,1.0],[1,5.2361],[2,5.7589],[3,5.6207],[4,5.4291],[5,5.2833],[6,5.1843]]

(%i145) wxdraw2d (xlabel = "t", yrange = [0,10], xrange = [0, 6],
title = "g = 1/2, j = 1/3, β = 2, k = 5/2, μ=2, α - T = 11",
key = "U[t]", points (tUL), color = magenta, key = "pi[t]",
points (tπL), color = black, line_width = 1, key = "Ue = 5",
explicit (5, t, 0, 6), color = brown, key = "μ = 2", explicit (2, t, 0, 6))$

(%t145)



7.1.3 Using DE2S (A1, A2, B, t, Y0)

(%i146) Ys : DE2S (A1_case1, A2_case1, B_case1, t, Y0);

(Ys)

$$\begin{pmatrix} \frac{(\sqrt{241}+31)^t\,(125\sqrt{241}+241)}{241\ 72^t}-\frac{(31-\sqrt{241})^t\,(125\sqrt{241}-241)}{241\ 72^t}+2 \\ -\frac{(\sqrt{241}-29)\,(\sqrt{241}+31)^t\,(125\sqrt{241}+241)}{11568\ 72^t}-\frac{(31-\sqrt{241})^t\,(\sqrt{241}+29)\,(125\sqrt{241}-241)}{11568\ 72^t}+5 \end{pmatrix}$$

(%i147) [πs, Us] : Ime (Ys),numer;

(%o147) $$\left[\frac{9.052\ 46.524^{t}}{72^{t}}-\frac{7.052\ 15.476^{t}}{72^{t}}+2,\frac{2.5413\ 46.524^{t}}{72^{t}}-\frac{6.5413\ 15.476^{t}}{72^{t}}+5\right]$$

(%i148) tπL : makelist ([tt, at (πs, t = tt)], tt, 0, 6);

(tπL) [[0,4.0],[1,6.3333],[2,5.4537],[3,4.3722],[4,3.563],[5,3.0165],[6,2.6582]]

(%i149) tUL : makelist ([tt, at (Us, t = tt)], tt, 0, 6);

(tUL) [[0,1.0],[1,5.2361],[2,5.7589],[3,5.6207],[4,5.4291],[5,5.2833],[6,5.1843]]

(%i150) wxdraw2d (xlabel = "t", yrange = [0,10], xrange = [0, 6],
title = "g = 1/2, j = 1/3, β = 2, k = 5/2, μ=2, α - T = 11",
key = "U[t]", points (tUL), color = magenta, key = "pi[t]",
points (tπL), color = black, line_width = 1, key = "Ue = 5",
explicit (5, t, 0, 6), color = brown, key = "μ = 2", explicit (2, t, 0, 6))$

(%t150)



7.1.4 Changing Y0

We continue with the same case1 parameters but change the initial conditions, using Ytlist for simplicity, first with π[0] = 4, U[0] = 7.

(%i151) Y0 : cvec ([4, 7]);

(Y0) $$\begin{pmatrix}4\\7\end{pmatrix}$$

(%i152) Ys : Ytlist (D, E, Y0, 6), numer;

(Ys) $\begin{pmatrix} [4,2.3333,2.0093,1.9617,1.9657,1.9758,1.9839] \\ [7,5.4028,5.0691,5.0035,4.9934,4.9939,4.9956] \end{pmatrix}$

(%i153) [πs, Us] : lme (Ys);

(%o153) [[4,2.3333,2.0093,1.9617,1.9657,1.9758,1.9839],[7,5.4028,5.0691,5.0035,4.9934,4.9939,4.9956]]

We need the list of times (tL) to be the same length as πs and Us.

(%i154) tL : makelist (tt, tt, 0, 6);

(tL) [0,1,2,3,4,5,6]

(%i155) wxdraw2d (xlabel = "t", yrange = [0,10], xrange = [0, 6],
title = "g = 1/2, j = 1/3, β = 2, k = 5/2, μ=2, α - T = 11",
key = "U[t]", points (tL, Us), color = magenta, key = "pi[t]",
points (tL, πs), color = black, line_width = 1, key = "Ue = 5",
explicit (5, t, 0, 6), color = brown, key = "μ = 2", explicit (2, t, 0, 6))$

(%t155)


We see both inflation and unemployment decreasing smoothly to their long term values.

We next start with U[0] = 7, π[0] = 0.5.

(%i156) Y0 : cvec ([0.5, 7]);

(Y0) $\begin{pmatrix} 0.5 \\ 7 \end{pmatrix}$

```
(%i157) Ys : Ytlist (D, E, Y0, 6), numer;
(Ys)    [ [0.5,−0.58333,−0.016204,0.62262,1.094,1.4111,1.6187] ]
        [ [7,4.7951,4.5458,4.6373,4.7508,4.8358,4.8932]       ]

(%i158) [πs, Us] : lme (Ys);
(%o158) [[0.5,−0.58333,−0.016204,0.62262,1.094,1.4111,1.6187],[7,
        4.7951,4.5458,4.6373,4.7508,4.8358,4.8932]]
```

We need the list of times (tL) to be the same length as πs and Us.

```
(%i159) tL : makelist (tt, tt, 0, 6);
(tL)    [0,1,2,3,4,5,6]

(%i160) wxdraw2d (xlabel = "t", yrange = [0,10], xrange = [0, 6],
         title = "g = 1/2, j = 1/3, β = 2, k = 5/2, μ=2, α - T = 11",
         key = "U[t]", points (tL, Us), color = magenta, key = "pi[t]",
         points (tL, πs), color = black, line_width = 1, key = "Ue = 5",
         explicit (5, t, 0, 6), color = brown, key = "μ = 2", explicit (2, t, 0, 6))$
```

(%t160) 

We see some oscillation of unemployment around Ue = 5 but eventual convergence of both U[t] and π[t].

7.2 Case 2: Complex Eigenvalues

```
(%i161) case2 : [β = 2, k = 5/2, g = 0.9, j = 0.9, μ = 2];
```

(case2) $[\beta=2, k=\frac{5}{2}, g=0.9, j=0.9, \mu=2]$

```
(%i165) A1_case2 : at (A1, case2);
        A2_case2 : at (A2, case2);
        B_case2 : at (B, case2);
        B_case2 : ratsubst (11, α - T, B_case2);
```

(A1_case2) $\begin{pmatrix} 1 & 0 \\ -2.25 & 6 \end{pmatrix}$

(A2_case2) $\begin{pmatrix} 0.91 & -1.8 \\ 0 & 1 \end{pmatrix}$

(B_case2) $\begin{pmatrix} 0.9\,(\alpha - T) \\ \frac{5\,(\alpha - T - 2)}{2} \end{pmatrix}$

(B_case2) $\begin{pmatrix} \frac{99}{10} \\ \frac{45}{2} \end{pmatrix}$

Redefine D and E for case2.

```
(%i167) D : invert (A1_case2) . A2_case2;
        E : invert (A1_case2) . B_case2;
```

(D) $\begin{pmatrix} 0.91 & -1.8 \\ 0.34125 & -0.50833 \end{pmatrix}$

(E) $\begin{pmatrix} 9.9 \\ 7.4625 \end{pmatrix}$

```
(%i168) [r1, r2] : eigenvalues (D)[1];
```

(%o168) $[-\frac{\sqrt{160319}\,\%i - 241}{1200}, \frac{\sqrt{160319}\,\%i + 241}{1200}]$

```
(%i169) abs(r1), numer;
```

(%o169) 0.38944

We have complex eigenvalues rather than real as in case1. r1 and r2 are complex conjugates of each other. For convergence we need the absolute value of either root to be less than 1. Since |r1| = 0.4 < 1, we have a convergent case.

```
(%i170) Ye : invert (ident(2) - D) . E;
```

(Ye) $\begin{pmatrix} 2.0 \\ 5.4 \end{pmatrix}$

We expect convergence to the values πe = 2 = μ and Ue = 5.4.

For our first plot with case2, we assume π[0] = 4 and U[0] = 1.

(%i171) Y0 : cvec ([4, 1]);

(Y0) $\begin{pmatrix} 4 \\ 1 \end{pmatrix}$

(%i172) Ys : Ytlist (D, E, Y0, 6), numer;

(Ys) $\begin{pmatrix} [4,11.74,5.6089,1.9723,1.4415,1.7799,1.9963] \\ [1,8.3192,7.2399,5.6963,5.24,5.2908,5.3804] \end{pmatrix}$

(%i173) [πs, Us] : lme (Ys);

(%o173) [[4,11.74,5.6089,1.9723,1.4415,1.7799,1.9963],[1,8.3192,7.2399,5.6963,5.24,5.2908,5.3804]]

We need the list of times (tL) to be the same length as πs and Us.

(%i174) tL : makelist (tt, tt, 0, 6);

(tL) [0,1,2,3,4,5,6]

(%i175) wxdraw2d (xlabel = "t", yrange = [0,10], xrange = [0, 6],
title = "g = 1/2, j = 1/3, β = 2, k = 5/2, μ=2, α - T = 11",
key = "U[t]", points (tL, Us), color = magenta, key = "pi[t]",
points (tL, πs), color = black, line_width = 1, key = "Ue = 5.4",
explicit (5.4, t, 0, 6), color = brown, key = "μ = 2", explicit (2, t, 0, 6))$

(%t175)



Both π[t] and U[t] undergo one large oscillation before settling down.

Changing Y0 in case 2, with π[0] = 4, U[0] = 7.

(%i176) Y0 : cvec ([4, 7]);

(Y0) $\begin{pmatrix} 4 \\ 7 \end{pmatrix}$

(%i177) Ys : Ytlist (D, E, Y0, 6), numer;

(Ys) $\begin{pmatrix} [4,0.94,1.2709,1.8679,2.0575,2.0431,2.0086] \\ [7,5.2692,5.1048,5.3013,5.4051,5.417,5.4061] \end{pmatrix}$

(%i178) [πs, Us] : lme (Ys);

(%o178) [[4,0.94,1.2709,1.8679,2.0575,2.0431,2.0086],[7,5.2692,5.1048,5.3013,5.4051,5.417,5.4061]]

```
(%i179) wxdraw2d (xlabel = "t", yrange = [0,10], xrange = [0, 6],
 title = "g = 1/2, j = 1/3, β = 2, k = 5/2, μ=2, α - T = 11",
 key = "U[t]", points (tL, Us), color = magenta, key = "pi[t]",
 points (tL, πs), color = black, line_width = 1, key = "Ue = 5.4",
 explicit (5.4, t, 0, 6), color = brown, key = "μ = 2", explicit (2, t, 0, 6))$
```

(%t179)



After one oscillation, both π[t] and U[t] converge to their respective long term values.